Proof of Proposition 3.4

In[1]:=
```
SetDirectory["~/writing/WIP/KappaLib/"];
<< kappaLib10.m
```

```
KappaLib v1.0
```

Suppose g is a Riemann metric of arbitrary signature and kappa is the Hodge operator. We show that the Tamm-Rubilar tensor density satisfies

G^abcd xi_a xi_b xi_c xi_d = sgn(det g) sqrt(abs(det(g))) (g(xi,xi))^2

In[3]:=
$$\text{invMetric} = \begin{pmatrix} h11 & h12 & h13 & h14 \\ h12 & h22 & h23 & h24 \\ h13 & h23 & h33 & h34 \\ h14 & h24 & h34 & h44 \end{pmatrix};$$

In[4]:=
```
Union[Flatten[invMetric - Transpose[invMetric]]]
```

Out[4]=
```
{0}
```

In[5]:=
```
(*
 *  The below is adopted from emHodge with the difference that the below
 *  computes the Hodge star operator as a function of sqrt(abs(det g))
 *  and g^-1.
 *
 *)
emHodgeLow[detSq_, invMetric_] := Module[
    {resKappa,  i, j , m, n, a, b},

    resKappa = emZeroKappa[];

    For[i = 1, i <= 4, i++,
        For[j = i + 1, j <= 4, j++,
            For[m = 1, m <= 4, m++,
                For[n = m + 1, n <= 4, n++,
                    resKappa[[i]][[j]][[m]][[n]] = detSq Sum[

                  invMetric[[i]][[a]] invMetric[[j]][[b]] Signature[{a, b, m, n}],
                                  {a, 1, 4}, {b, 1, 4}
                              ];
                 ];
             ];
         ];
    ];

    resKappa

]
```

In[6]:=
```
(* Compute (2,2)-tensor induced by the above metric tensor *)
kappa = emHodgeLow[detSq, invMetric];
```

In[7]:=
```
(* Compute and simplify Fresnel equation *)
xi = {xi0, xi1, xi2, xi3};
fresnel = emKappaToFresnel[kappa, xi];
FullSimplify[fresnel]
```

Out[9]=
$$-\text{detSq}^3 \left(-h13^2\, h24^2 + h11\, h24^2\, h33 + h14^2 \left(-h23^2 + h22\, h33\right) + 2\, h12\, h13\, h24\, h34 - 2\, h11\, h23\, h24\, h34 - h12^2\, h34^2 + h11\, h22\, h34^2 + 2\, h14\, (h13\, h23\, h24 - h12\, h24\, h33 - h13\, h22\, h34 + h12\, h23\, h34) + \left(h13^2\, h22 - 2\, h12\, h13\, h23 + h11\, h23^2 + h12^2\, h33 - h11\, h22\, h33\right) h44\right) \left(h11\, xi0^2 + 2\, h12\, xi0\, xi1 + h22\, xi1^2 + 2\, h13\, xi0\, xi2 + 2\, h23\, xi1\, xi2 + h33\, xi2^2 + 2\, (h14\, xi0 + h24\, xi1 + h34\, xi2)\, xi3 + h44\, xi3^2\right)^2$$

In[10]:=
```
fresnelClosedForm = detSq^3 Det[invMetric] (xi.invMetric.xi)^2;
```

In[11]:=
```
Simplify[fresnel - fresnelClosedForm]
```

Out[11]=
```
0
```

■ Check that trace kappa = 0

In[13]:=
```
Simplify[emTrace[kappa]]
```

Out[13]= 0